

Julia User Group Meeting

GUIs with GLMakie

Boris Kaus


Graphical User Interfaces using GLMakie

First GUI

- Let's start with a very simple plot:

```
# Example 1 - simple bar plot
x = [1,2,3]
y = [2,1,3]

using GLMakie
fig = Figure()
ax = Axis(fig[1,1])
barplot!(ax, x, y)
display(fig)
```




Observable

- If you make a variable an **Observable**, it will update the plot once this variable is changed

```
x = [1,2,3]
y = Observable([2,1,3])
```

```
using GLMakie
fig = Figure()
ax = Axis(fig[1,1])
barplot!(ax, x, y)
display(fig)
```

```
y[] = [1,5,4] # update y
display(fig) # show figure again
```




Let's add a Button

- You can adjust the location by making `fig[1,1:3]` larger:

```
x = [1,2,3]
y = Observable([2,1,3])

using GLMakie
fig = Figure()
ax = Axis(fig[1,1:3])
but = Button(fig[2,2], label="y[2]")
barplot!(ax, x, y)

display(fig)
```




Let's add some action to the Button

- Use the `on(...)` `do n` construct:

```
x = [1,2,3]
y = Observable([2,1,3])

using GLMakie
fig = Figure()
ax = Axis(fig[1,1:3])
but = Button(fig[2,2], label="y[2]")
barplot!(ax, x, y)


on(but.clicks) do n
    @show n
    y[][2] += 1      # update y[2]
    notify(y)        # update plot
end
display(fig)
```



```
julia> n = 1
n = 2
n = 3
```

Exercise 1

1. Add 2 additional buttons with which you can increase $y[1]$ and $y[3]$
2. Add another row of buttons with which you can decrease the values by 1



Different GUI items

Colormap

viridis ▼

Menu


Count: 4

Button

☒ Dataset A

☐ Dataset B

Checkbox



Slider


Live Update



Toggle



Box



Intervalslider

Enter a string...

Click to edit...

Textbox

For help see the “blocks” part of the documentation:
<https://docs.makie.org/stable/reference/blocks/button>


Exercise 2 – more sophisticated GUI

- Create a GUI that plots either one of the two functions:
 - $f(x) = a \cdot x^3 + b \cdot x^2$
 - $f(x) = a \cdot x^2 + b \cdot x$
 - Over the interval from $x=-2:.01:2$
- Features of the GUI:
 - Dropdown menu to select the function
 - Dropdown menu to select the color of the line (`:red`, `:blue`, `:green`)
 - Textboxes to specify `a` and `b`

Feedback after clicking on an axis

- You can add feedback when you click on an axis
- Example:
 - Use your mouse to select a point
 - Use this to select a river in a catchment area

```
Location of Point changed to [0.8830
Location of Point changed to [0.8830
Location of Point changed to [0.8830
Location of Point changed to [0.1031
Location of Point changed to [0.1031
Location of Point changed to [0.1031
Location of Point changed to [0.1031
Location of Point changed to [0.8601
Location of Point changed to [0.8601
Location of Point changed to [0.8601
Location of Point changed to [0.8601711392402649, 0.20663413405418396]
```



```
# Click somewhere in the axis
fig = Figure()
ax = Axis(fig[1, 1])

# switch off zoom etc
deactivate_interaction!(ax, :scrollzoom)
deactivate_interaction!(ax, :limitreset)
deactivate_interaction!(ax, :dragpan)
deactivate_interaction!(ax, :rectanglezoom)

# Create a horizontal & vertical line
vlines!(ax, 0.5, linewidth=10)
hlines!(ax, 0.5, linewidth=10)

# add a point
point = Observable(Point2{Float64}(0.5, 0.5))
scatter!(ax, point, color=:red)


# Change coordinates of point on mouse click
on(events(ax.scene).mousebutton) do e
    position = Float64.(mouseposition(ax.scene))
    point[] = position # update point on plot
    notify(point)
end

# Write something if point changes
on(point) do p
    println("Location of Point changed to $p")
end
fig
```

Draw a curve interactively


```
# Draw a line with mouse clicks
fig = Figure()
ax = Axis(fig[1, 1])
for interact in keys(ax.interactions)
    deactivate_interaction!(ax, interact)
end
points = Observable{Point2f{}}()
linesegments!(ax, points, color = :black)
scatter!(ax, points, color = :gray)

on(events(ax.scene).mousebutton) do event
    if event.button == Mouse.left
        if event.action == Mouse.press || event.action == Mouse.release
            mp = mouseposition(ax.scene)
            push!(points[], mp)
            notify(points)
        end
    end
end
fig
```



Other packages – Dash.jl

- Run GUI's directly in the browser
- More complicated to develop
- e.g., **MAGEMinApp.jl** or **InteractiveGeodynamics.jl**



Exercise 3 - topography

- Create a GUI that uses the GeophysicalModelGenerator package to plot a topography from a certain region area
- Specify lower-left and upper-right longitude-latitude corners

```
# GUI to plot Topography
using GeophysicalModelGenerator, GMT, GLMakie

# Example of loading Topo:
Topo = import_topo([4,20,37,49]);

# Plot Topo with heatmap(Topo) or with:
heatmap(Topo.lon.val[:,1,1],Topo.lat.val[1,:,1], ustrip.(Topo.fields.Topography[:, :, ]))
```

Summary

- You made your first GUIs